
1 はじめに

前回から連載の「はじめての Fortran90」シリー

ズの二回目です。

前回は、 Fortran90の新しい機能について述べ、

プログラムの形式として自由形式、データ型の宣

言、手続き、配列演算機能について解説しました。

今回は、

� 再帰手続き

� 構造体

� ポインタ

� 構造型とポインタを用いた動的なデータ構造

について解説します。

2 再帰手続き

再帰手続きとは、ある手続きが直接あるいは間接

的に自分自身を呼び出すような構造になっている手

続きをいいます。

2.1 再帰手続きの宣言

FORTRAN77では再帰手続きを書くことは許さ

れていませんでしたが、 Fortran90から次のよう

に FUNCTION文、 SUBROUTINE文の前にキー

ワードRECURSIVEをつけることで再帰手続

きが宣言できます。

RECURSIVE SUBROUTINE name(arg,..)
：

END SUBROUTINE name

RECURSIVE FUNCTION name(arg,..) &
[RESULT(name)]

：
END FUNCTION name

2.2 再帰関数とRESULT句

FORTRAN77の関数宣言では、結果を返す式は

関数名への代入式として表現していました (図 1)。

REAL FUNCTION add(arg1,arg2)
IMPLICIT NONE
REAL, INTENT(IN) :: arg1,arg2
add = arg1+arg2

END FUNCTION add

図 1. RESULT句の無い関数

しかし、これは手続き名への代入というプログラ

ムは読みにくものです。 Fortran90ではRESULT

句で関数の結果を返す変数名を指定できます (図

2)。

FUNCTION add(arg1,arg2) RESULT(result)
IMPLICIT NONE
REAL,INTENT(IN) :: arg1,arg2
REAL :: result
result = arg1+arg2

END FUNCTION add

図 2. RESULT句を指定した関数

ただし、RESULT句を指定した関数は、 FUNC-

TION文にデータ型は指定できません。また、直

接自分自身を呼び出す再帰関数の宣言ではRE-

SULT句の指定が必須になります。

入力された数の階乗を求めるプログラム factrial

を再帰関数として宣言すると図 3のようになりま

す。

PROGRAM fact
IMPLICIT NONE
INTEGER :: n
READ(*,*) n
WRITE(*,*) n,factrial(n)
CONTAINS
RECURSIVE FUNCTION factrial(n) &

RESULT(result)
INTEGER,INTENT(IN) :: n
INTEGER :: result
IF ( n > 0 ) THEN
result = n*factrial(n-1)
ELSE
result = 1
END IF
END FUNCTION factrial

END PROGRAM fact

図 3. 再帰関数の例



2.3 再帰サブルーチン

階乗プログラムのサブルーチン版を図 4に示しま

す。

PROGRAM fact
IMPLICIT NONE
INTEGER :: n,result
READ(*,*) n
CALL fact(n,result)
WRITE(*,*) n,result
CONTAINS
RECURSIVE SUBROUTINE factrial(n,result)
INTEGER,INTENT(IN) :: n
INTEGER,INTENT(INOUT) :: result
IF ( n > 0 ) THEN
CALL factrial(n-1,result)
result = n*result
ELSE
result = 1
END IF

END SUBROUTINE factrial
END PROGRAM fact

図 4. 再帰サブルーチンの例

2.4 再帰手続き VS 繰返し

アルゴリズムには、再帰的なプログラム表現にな

るものが多くあり、簡潔に表現できるので再帰手

続きが宣言できることは大変便利です。しかし、計

算機にとって手続きの呼出し処理は多くのコストを

要するものです。したがって、高速な演算のために

は、再帰的な処理を繰返しに置き換えることも必要

です。図 5に階乗プログラムのDOループ版を示し

ます。

PROGRAM fact
IMPLICIT NONE
INTEGER :: n,result
READ(*,*) n
CALL factrial(n,result)
WRITE(*,*) n,result
CONTAINS
SUBROUTINE factrial(n,result)
INTEGER,INTENT(IN) :: n
INTEGER,INTENT(INOUT) :: result
INTEGER :: i
result = 1
DO i = 1, n
result = result*i
END DO

END SUBROUTINE factrial
END PROGRAM fact

図 5. 階乗の繰返しプログラム例

3 構造体

Fortranには、組込みデータ型として整数型、実

数型、複素数型、論理型、文字型の五つがありま

す。 Fortrn90では、この五つの組込みデータ型に

加えて、利用者が新たにデータ型を定義し、拡張す

る枠組みを備えています。この利用者定義のデータ

型を構造型といい、構造型で定義された変数を構造

体といいます。

3.1 構造型の定義と宣言

構造体を宣言するには、まず、構造型の定義が必

要です。構造型定義の形式は、次のようになってい

ます。

TYPE derived_type_name
component_define

END TYPE derived_type_name

ここで、 derived type nameは、利用者定義

の構造型の名前です。また、 component de�ne

には、定義する構造型の成分をデータ型の宣言文と

同じ形式で並べたものを指定します。構造型の成分

には、五つの組込みデータ型とそれ自身を含む構造

型を宣言することができます。

例えば、 X-Y座標系での一点を表す構造型 point

の定義は、次のようになります。

TYPE point
REAL :: x,y

END TYPE point

ここで、構造型 pointは、 x,yという実数型の二

つの成分を持つデータ型として定義されます。一

方、構造型を用いた構造体宣言の形式は、次のよう

に指定します。

TYPE(derived_type_name)[,attribute] &
:: name,...

したがって、構造型 pointの構造体 p1,p2の宣

言は、次のようになります。

TYPE point
REAL :: x,y

END TYPE point
：

TYPE(point) :: p1,p2

3.2 構造体の演算と成分構成子 (%)

利用者が宣言した構造体を直接演算式に用いるこ

とはできません。構造体の演算式は、構造型の個々

の成分を指定した演算として記述する必要があり



ます。構造体の各成分を指定するには、成分構成

子%(パーセント)を用います。前に宣言した構造

体変数 p1,p2の各成分は、次のように表します。

点 p1の x座標 p1%x

点 p1の y座標 p1%y

点 p2の x座標 p2%x

点 p2の y座標 p2%y

また、演算式は次のように表します。

p1%y=1.0

p1%x=p1%x+p2%x

x=p1%x*p2%x

ここで、 xは実数型の変数です。

3.3 構造体の代入と構造型の定数

構造体の演算式は、個々の成分を指定しないと表

せませんが、同じ構造型を持つ構造体変数同士であ

れば代入できます。

p1=p2 !(1)

p1%x=p2%x; p1%y=p2%y !(2)

この式 (1)と式 (2)は同じ処理を表し、構造型の

対応する成分がそれぞれ代入されます。また、構造

型の定数は、次のような形式で表します。

derived_type_name(value,...)

すなわち、先頭に構造型名を指定し、各成分の与

える定数を順に,(カンマ)で区切り、全体を括弧で

囲んで表します。これを構造体構成子といいます。

p1%x=3.0; p1%y=7.0 !(1)

p1=point(3.0,7.0) !(2)

式 (1)の成分構成子を使った代入式は、式 (2)の

ように構造体構成子を用いて表します。また、構造

体構成子は構造体の宣言で初期値として指定できま

す。

3.4 構造体の入出力

入出力文に構造体変数名を直接書くことができま

す。この場合構造型の定義で書いた順に各成分が処

理されます。すなわち、 次の式 (1)と式 (2)の結

果は同じになります。

WRITE(*,*) p1,p2 !(1)

WRITE(*,*) p1%x,p1%y,p2%x,p2%y !(2)

3.5 構造体を用いたプログラム例

構造体を用いたプログラム例として X-Y座標上

の二点 p1と p2を与え、その二点と原点 (0,0)と

で囲まれる三角形の面積を求めるプログラムを図

6に示します。

PROGRAM s_triangle
IMPLICIT NONE
TYPE point
REAL :: x,y
END TYPE point
TYPE(point) :: p1,p2
REAL :: s
READ(*,*) p1,p2
s=ABS(p1%x*p2%y-p1%y*p2%x)/2.0
WRITE(*,*) s

END PROGRAM s_triangle

図 6. 構造体を用いたプログラム例

3.6 色々な構造体の例

1) 構造体を成分に含む構造体

図 7は、 X-Y座標の一点を表す pointという構

造型を定義し、三角形の各頂点 a,b,cを成分とす

る構造型 triangleの定義と構造体変数 tの宣言で

す。

TYPE point
real :: x,y

END TYPE point
TYPE triangle

TYPE(point) :: a,b,c
END TYPE triangle
TYPE(triangle) :: t

図 7. 構造体を成分に含む構造体の例

この場合、 triangle型の変数 tの成分 a,b,cは、

それぞれ構造型 pointであるために演算式では、

t%a%x,t%a%yように表します。

t%a%xのように示すものを構造体の末端成分と

いい、演算式は末端成分で表します。また、この構

造型 triangleの構造体構成子は、次のように表し

ます。

t=triangle(point(0.0,0.0), &
point(1.0,4.0),point(4.0,1.0))



2) 構造型の配列

図 8のプログラムの構造体 p1,p2は、 100個の

構造型 pointを要素とする配列の宣言です。このプ

ログラムは、 p1,p2の対になったデータを読み、

それぞれの面積を求めています。

PROGRAM s_triangle
IMPLICIT NONE
INTEGER, PARAMETER :: n=100
TYPE point
REAL :: x,y
END TYPE point
TYPE(point),DIMENSION(n) :: p1,p2
REAL,DIMENSION(n) :: s
INTEGER :: i
DO i=1,n
READ(*,*) &
p1(i)%x,p1(i)%y,p2(i)%x,p2(i)%y

END DO
s=ABS(p1%x*p2%y-p1%y*p2%x)/2.0
WRITE(*,*) s

END PROGRAM s_triangle

図 8. 構造型の配列

3) 配列を成分にもつ構造体

図 9プログラムの構造型 groupは、成分として

整数型の識別子 idとその値の持つ 100個の実数型

配列 valueから構成されています。このプログラム

は、 idがゼロになるまでデータを読み、 idとその

総和を出力します。

PROGRAM group_sum
IMPLICIT NONE
TYPE group
INTEGER :: id
REAL,DIMENSION(100) :: value
END TYPE group
TYPE(group) :: work
DO
READ(*,*) work
IF (work%id == 0) EXIT
WRITE(*,*) work%id,sum(work%value(:))
END DO

END PROGRAM group_sum

図 9. 配列を成分に含む構造体の例

4 ポインタ

ポインタとは、 Fortran90で新たに導入された

もので、図 10に示すように他の変数領域を指す変

数です。ポインタは型宣言文で POINTER属性を

指定して宣言します。

ポインタが指すことができる領域には、次の二種

類があります。

Pointer Variable area

図 10. ポインタの概念

1) 型宣言文で TAGET属性が指定された変数。

2) ポインタ変数を指定した ALLOCATE文で動

的に割付けられた領域。

4.1 ポインタの宣言

ポインタ変数とその指示先であるターゲット変数

は、それぞれ変数の型宣言でPOINTER属性、

TARGET属性を指定します。

REAL,POINTER :: p

REAL,TARGET :: t

ここで、

� ポインタ変数は、 TERGET属性を持たない

変数は指せません。

� ポインタ変数は、同じデータ型のターゲット

変数でないと指せません。

4.2 ポインタとターゲットの結合

宣言されたポインタ変数は、何も指示していない

状態であり、

p => t

のような式でターゲット変数と結合し、指示しない

限り演算式に指定できません。この式をポインタ代

入といい=>で表します。

4.3 ポインタの操作

4.3.1 ポインタ代入

図 11のプログラムは、式 (1)の操作でポインタ

変数 p1はターゲット変数 t1と結合し、式 (2)の

操作でポインタ変数 p2はターゲット変数 t2と結

合します。

式 (1),(2)の結果ポインタ変数 p1,p2は、図 12の

ようにそれぞれターゲット変数 t1,t2の領域を指し

ます。



PROGRAM pointer
IMPLICIT NONE
REAL,POINTER :: p1,p2
REAL,TARGET :: t1=1.0,t2=2.0
p1=>t1 !(1)
p2=>t2 !(2)
WRITE(*,*) p1,t1,p2,t2 !(3)

END PROGRAM pointer

図 11. プログラム例 1

t1

t2

1.0

2.0

図 12. ポインタ代入の結果

式 (3)のWRITE文の結果は、

1.0 1.0 2.0 2.0

となります。すなわち、ポインタ変数 p1,p2は、

ターゲット変数 t1,t2の別名として機能していま

す。

ここで、図 11のプログラムの後に、次のような

ポインタ変数に対する演算を加えると

p2=>t1 !(4)
WRITE(*,*) p1,t1,p2,t2 !(5)

式 (4)のポインタ代入の結果、ポインタ変数 p2の

指示先はターゲット変数 t1に変更され、 (5)のWRITE

文の結果は、

1.0 1.0 1.0 2.0

となります。

すなわち、式 (4)の実行後のポインタ変数 p2も

図 13のようにターゲット変数 t1の領域を指しま

す。

p1

p2

t1

t2

1.0

2.0

図 13. 指示先の変更

ここで、式 (4)のポインタ代入は

p2=>p1 !(4)
WRITE(*,*) p1,t1,p2,t2 !(5)

のように、ポインタ変数同士をポインタ代入しても

同じ結果になります。

4.3.2 ポインタ変数へ代入

これまでポインタ代入について説明しましたが、

次はポインタ変数への代入操作について説明しま

す。

図 14のプログラムは、図 11のプログラムとは

異なり、ターゲット変数は初期値を持っていませ

ん。式 (1)は、ポインタ変数 p1,p2とターゲット

変数 t1,t2をそれぞれ結合し、指示する操作です。

ここでターゲット変数 t1,t2の値は未定義であり、

t1,t2,p1,p2をWRITEしても結果は不定です。式

(2)で p1,p2に対する代入で指示先、 t1、 t2にそ

れぞれ 1.0,2.0が代入されます。

PROGRAM pointer
IMPLICIT NONE
REAL,POINTER :: p1,p2
REAL,TARGET :: t1,t2
p1=>t1 ; p2=>t1 !(1)
p1=1.0 ; p2=2.0 !(2)
WRITE(*,*) p1,t1,p2,t2 !(3)

END PROGRAM pointer

図 14. プログラム例 2

式 (3)のWRITE文の結果は、図 11のプログラ

ムと同じで、次のようになります。

1.0 1.0 2.0 2.0

ここで、図 14のプログラムの後に、次のような

ポインタ変数同士の代入加えると、

p2=p1 !(4)
WRITE(*,*) p1,t1,p2,t2 !(5)

式 (5)のWRITE文の結果は、次のようになりま

す。

1.0 1.0 1.0 1.0

すなわち、 p1の指示先の t1の値 1.0が p2の指

示先の t2に代入されます。 式 (4)の実行後の各変

数は、図 15のようになっています。

p1

p2

t1

t2

1.0

1.0

図 15. ポインタ変数同士の代入の結果



4.4 ポインタ配列

配列宣言でPOINTER属性を指定するとポイ

ンタ配列が宣言できます。多次元のポインタ配列

も宣言できますが、個々の次元の寸法を指定するこ

とはできません。多次元のポインタ配列は、配列の

次元寸法の代わりに:(コロン)を,(コンマ)で区切っ

て、次元数だけを指定します。例えば、一次元、二

次元のポインタ配列は、次のように宣言します。

REAL,POINTER,DIMENSION(:) :: ptr_1d

REAL,POINTER,DIMENSION(:,:) :: ptr_2d

ポインタ配列と結合する配列は、TERGRT属

性を持つ必要があります。

REAL,TARGET,DIMENSION(5) :: t_1d

REAL,TARGET,DIMENSION(5,5) :: t_2d

PROGRAM pointer_array
IMPLICIT NONE
INTEGER :: i
REAL,POINTER,DIMENSION(:) :: ptr_1d
REAL,POINTER,DIMENSION(:,:) :: ptr_2d
REAL,TARGET,DIMENSION(5,5) :: t_2d &
= RESHAPE((/ (i,i=1,25) /), (/5, 5/))
ptr_2d => t_2d !(1)
DO i=1,5

WRITE(*,*) (ptr_2d(i,:)) !(2)
END DO
ptr_1d => t_2d(:,1) !(3)
WRITE(*,*) ptr_1d !(4)
ptr_1d => t_2d(1,:) !(5)
WRITE(*,*) ptr_1d !(6)
ptr_2d => t_2d(1:2,1:2) !(7)
WRITE(*,*) ptr_2d !(8)
END PROGRAM pointer_array

図 16. ポインタ配列の結合

4.4.1 ポインタ配列の結合

ポインタ配列は、同じデータ型、同じ次元数を持

つターゲット配列とだけ結合し、ポインタ配列の各

要素がターゲット配列の各要素を指示します。ポイ

ンタ配列の各次元寸法は、結合したターゲット配列

の各次元寸法になります。

図 16のプログラムは、ポインタ配列が部分配列

式で指定されたターゲット配列と結合し、指示する

例です。式 (1)のポインタ代入は、ポインタ、ター

ゲットともに二次元の配列であるので、ptr_2dの

各要素がt_2dの各要素を指すように結合されま

す。式 (2)のWRITE文の結果は、次のようにな

ります。

1.0 6.0 11.0 16.0 21.0
2.0 7.0 12.0 17.0 22.0
3.0 8.0 13.0 18.0 23.0
4.0 9.0 14.0 19.0 24.0
5.0 10.0 15.0 20.0 25.0

式 (3)は、t_2dの一列目の全要素を一次元のポ

インタ配列ptr_1dで指しています。このようにポ

インタ代入文と部分配列式を用いると配列の一部を

指すことができます。式 (4)のWRITE文の結果

は、次のようになります。

1.0 2.0 3.0 4.0 5.0

式 (5)は、t_2dの一行目の全要素を一次元のポ

インタ配列ptr_1dで指しています。式 (6)のWRITE

文の結果は、次のようになります。

1.0 6.0 11.0 16.0 21.0

式 (7)は、配列t_2dのを特定の部分を二次元の

ポインタ配列ptr_2dで指しています。式 (8)のWRITE

文の結果は、次のようになります。

1.0 2.0 6.0 7.0

式 (7)の部分配列式の結果は、二次元の配列とな

るのでポインタ配列も二次元ptr_2dになります。

4.5 ポインタ引数

手続きの引数にポインタを指定することができま

す。呼出し元の実引数がポインタの場合、手続き側

の仮引数には、次の二種類があります。

� 仮引数もポインタ

� 仮引数はポインタ以外

4.5.1 仮引数はポインタ

仮引数をポインタとする場合、実引数もポインタ

であり、ポインタ配列の場合次元も一致しなければ

なりません。この場合手続きの呼出し時の結合状態

が手続きに渡されます。ただし、ポインタ仮引数に

は INTENT属性は指定できません。また、ポイ

ンタ仮引数を外部手続きで宣言する場合、呼出し元

に引用仕様の宣言が必須です。図 17にポインタを

仮引数とするプログラム例を示します。



PROGRAM pointer_arg1
IMPLICIT NONE
INTEGER :: i
REAL,POINTER,DIMENSION(:) :: p
REAL,TARGET,DIMENSION(5) :: t &
= (/ (i,i=1,5) /)
p => t
call sub(p)
WRITE(*,*) t
CONTAINS
SUBROUTINE sub(v)
REAL,POINTER,DIMENSION(:) :: v
v = v + 100.0

END SUBROUTINE sub
END PROGRAM pointer_arg1

図 17. ポインタ仮引数の例

4.5.2 仮引数はポインタ以外

ポインタ実引数は、ポインタ以外の仮引数と対応

することも可能です。この場合ポインタは結合して

いなければならず、指示先が仮引数に渡ります。図

18にポインタ以外を仮引数とするプログラム例を

示します。

PROGRAM pointer_arg2
IMPLICIT NONE
INTEGER :: i
REAL,POINTER,DIMENSION(:) :: p
REAL,TARGET,DIMENSION(5) :: t &
= (/ (i,i=1,5) /)
p => t
call sub(p)
WRITE(*,*) t
CONTAINS
SUBROUTINE sub(v)
REAL,DIMENSION(:),INTENT(INOUT) :: v
v = v + 100.0

END SUBROUTINE sub
END PROGRAM pointer_arg2

図 18. ポインタ以外の仮引数の例

4.6 ポインタの動的割付け

ポインタは、TARGET属性を持つ変数を指示

するだけでなく、ALLOCATE文にポインタを

指定すると動的に割付けた領域と結合できます。図

19のプログラムは、ポインタの動的割付けの例で

す。

図 18のプログラムは、式 (1)で入力するデータ

個数を読込み、式 (2)のALLOCATE文で必要

な大きさの領域を割付け、式 (3)でデータを読込

み、式 (4)で総和を取り、出力しています。式 (5)

のDEALLOCATE文は、ALLOCATE文で確

保した領域を解放します。

ここで、TERGET属性を持つ変数とポインタ

変数との結合は、ポインタ代入を用いて結合する必

PROGRAM alloc_pointer
IMPLICIT NONE
INTEGER :: n
REAL,POINTER,DIMENSION(:) :: p
READ(*,*) n !(1)
ALLOCATE(p(n)) !(2)
READ(*,*) p !(3)
WRITE(*,*) SUM(p) !(4)
DEALLOCATE(p) !(5)
END PROGRAM alloc_pointer

図 19. ポインタの動的割付け例

要がありましたが、式 (2)のポインタ配列を指定し

たALLOCATE文は、 n個の一次元の実数配列

領域を確保し個々の要素がポインタ配列 pと結合

していることに注意してください。また、式 (5)の

DEALLOCATE文で領域が解放され、ポインタ

配列 pの結合も解かれます。

4.7 動的割付け配列 VS ポインタ配列

図 19のプログラムは、ALLOCATABLE属性

を指定した動的割付け配列を用いて表せます (図

20)。

PROGRAM alloc_array
IMPLICIT NONE
INTEGER :: n
REAL,ALLOCATABLE,DIMENSION(:) :: a
READ(*,*) n !(1)
ALLOCATE(a(n)) !(2)
READ(*,*) a !(3)
WRITE(*,*) SUM(a) !(4)
DEALLOCATE(a) !(5)
END PROGRAM alloc_array

図 20. 割付け配列の例

動的割付け配列とポインタ動的割付けの違いは、

動的割付け配列が名前を持った領域を確保し、効率

の良いプログラムを実現するのに対して、ポインタ

動的割付けは、無名の領域を割付け、線形リストな

ど柔軟な動的データ構造を持つプログラムを表現で

きることです。

4.8 ポインタの結合状態と文

ポインタ割付けを使った動的なデータ構造を説

明する前に、ポインタの結合状態とポインタの状

態に密接に関連するALLOCATE文、DEAL-

LOCATE文、NULLIFY文および状態を調べ

るためのASSOCIATED組込み関数の機能につ

いて説明します。



ポインタは、基本的に次の三つの状態を持ちま

す。

1) 未定義状態

未定義状態とは、ポインタが宣言されいるだけで

何とも結合せず、指示してない状態をいいます。

未定義状態のポインタをプログラムで参照するこ

とはできません。

2) 結合状態

結合状態とは、そのポインタが指示している状態

をいいます。

未定義状態のポインタが結合状態になるのは、ポ

インタ代入でTARGET属性を持つ変数と結合し

指示した場合と、ALLOCATE文で割付け領域

と結合した場合です。

3) 空状態

空状態とは、このポインタ変数が何も指示してい

ない状態をいいます。

NULLIFY文に指定されたポインタ変数は、空

状態になります。なお、ALLOCATE文で割付

けられたポインタ変数は、DEALLOCATE文

で解放すると、指定されたポインタも空状態になり

ます。

4.8.1 ALLOCATE文

ALLOCATE文にポインタを指定した場合、指

定された大きさの領域を割付け、ポインタがその領

域と結合されます。

ALLOCATE文は、次のような形式であり、複

数のポインタ変数が指定できます。 statは整変数

で割付けが成功した場合、ゼロが設定されます。

ALLOCATE(ptr_var,...[,STAT=stat])

ただし、 ALLOCATE文はポインタ変数の領域

を割付け、結合して返すだけです。

したがって、次のプログラムのように既に割付け

られたものと同じポインタ変数を ALLOCATE文

に指定すると、前の領域は他のポインタから指示さ

れていない限り参照不能になるので注意が必要で

す。

PROGRAM alloc_pointer
IMPLICIT NONE
REAL,POINTER :: p1,p2
ALLOCATE(p1)
p1=1.0

p2=>p1
ALLOCATE(p1)
p1=2.0
WRITE(*,*) p1,p2
END PROGRAM alloc_pointer

4.8.2 DEALLOCATE文

DEALLOCATE文は、ポインタ指示先を解放

し、ポインタを空状態にします。

DEALLOCATE文は、次のような形式であり、

複数のポインタ変数が指定できます。 statは整変

数であり解放が成功した場合、ゼロが設定されま

す。

DEALLOCATE(ptr_var,...[,STAT=sta])

ただし、DEALLOCATE文は指定されたポイ

ンタ変数の指示の領域を解放し、ポインタ変数を空

状態にするだけです。

したがって、次のプログラムのように他のポイン

タが指示する領域を DEALLOCATE文で解放す

るとそのポインタの状態は不定 (未定義)になって

しまいます。

PROGRAM dealloc_pointer
IMPLICIT NONE
REAL,POINTER :: p1,p2
ALLOCATE(p1)
p1=1.0
p2=>p1
DEALLOCATE(p1)
END PROGRAM dealloc_pointer

4.8.3 NULLIFY文

NULLIFY文は、次のような形式であり、指定

されたポインタ変数を空状態にします。

NULLIFY(ptr_var,...)

また、空状態のポインタ変数を他のポインタ変数

に、ポインタ代入するとそのポインタ変数も空状態

になります。

4.8.4 組込み関数ASSOCIATED

ポインタ変数の結合状態は、組込み関数 ASSO-

CIATEで知ることができます。

組込み関数 ASSOCIATEDは、次のような形式

で指定されたポインタ変数の結合状態を調べ、結果

を論理値で返します。



ASSOCIATED(pointer [,target])

引数に未定義なポインタ変数を指定することはで

きません。また、引数の指定により次のような結果

を返します。

1) targetを省略した場合

� pointerが結合状態のときは真 (.TRUE.)

� pointerが空状態のときは偽 (.FALSE.)

2) targetがTARGET属性を持つ変数の場合

� pointerが targetを指示していれば真 (.TRUE.)

� 上記以外は偽 (.FALSE.)

3) targetがPOINTER属性を持つ変数場合

� pointerが targetを指示していれば真 (.TRUE.)

� 上記以外は偽 (.FALSE.)

また、引数 pointer,targetのいずれかが空状態

の場合、ASSOCIATEは偽 (.FALSE.)を返します。

PROGRAM pointer_list
IMPLICIT NONE
TYPE list
INTEGER :: val
TYPE(list),POINTER :: next
END TYPE list
TYPE(list), POINTER :: root,work
INTEGER :: data
ALLOCATE(root)
NULLIFY(root%next)
work => root
create_list: DO
READ(*,*) data
IF ( data /= 0 ) THEN
work%val = data
ALLOCATE(work%next)
work=>work%next
NULLIFY(work%next)
ELSE
EXIT create_list
END IF
END DO create_list
work => root
print_list: DO
IF (ASSOCIATED(work%next)) THEN
WRITE(*,*) work%val
work=>work%next
ELSE
EXIT print_list
END IF
END DO print_list
work => root
free_list: DO
IF (ASSOCIATED(work%next)) THEN
root => work
work=>work%next
DEALOCATE(root)
ELSE
EXIT free_list
END IF
END DO free_list

END PROGRAM pointer_list

図 21. リスト構造プログラム

4.9 リスト構造プログラム

図 21に、ポインタの動的割付けの例として線形

リスト構造プログラムを示します。

4.10 木構造プログラム

図 22に、ポインタの動的割り付けの例として木

構造リストプログラムを示します。

PROGRAM tree_list
IMPLICIT NONE
TYPE node
INTEGER :: val
TYPE(node),POINTER :: left,right
END TYPE node
TYPE(node), POINTER :: tree
INTEGER :: data
NULLIFY(tree)
loop: DO
READ(*,*) data
IF ( data == 0 ) EXIT loop
CALL insert(tree,data)
END DO loop
CALL print_tree(tree)
CONTAINS
RECURSIVE SUBROUTINE insert(t, data)
TYPE(node), pointer :: t
INTEGER, INTENT(IN) :: data
IF(.not. ASSOCIATED(t)) THEN
ALLOCATE(t)
t%val=data
NULLIFY(t%left)
NULLIFY(t%right)
ELSE IF (data < t%val) THEN
CALL insert(t%left, data)
ELSE
CALL insert(t%right, data)
END IF
END SUBROUTINE insert
RECURSIVE SUBROUTINE print_tree(t)
TYPE(node), pointer :: t
IF(ASSOCIATED(t)) THEN
CALL print_tree(t%left)
WRITE(*,*) t%val
CALL print_tree(t%right)
END IF
END SUBROUTINE print_tree

END PROGRAM tree_list

図 22. 木構造プログラム

5 おわりに

連載記事「はじめての Fortran90」シリーズの二

回目です。今回は、再帰手続き、利用者定義のデー

タ型、構造型、ポインタについて解説しました。前

号でお約束していたモジュールの解説まで、たどり

着けませんでした。

次回は、文字列処理、入出力、モジュールを取り

上げたいと考えています。それではまた。


